Copyright>        OpenRadioss
Copyright>        Copyright (C) 1986-2023 Altair Engineering Inc.
Copyright>
Copyright>        This program is free software: you can redistribute it and/or modify
Copyright>        it under the terms of the GNU Affero General Public License as published by
Copyright>        the Free Software Foundation, either version 3 of the License, or
Copyright>        (at your option) any later version.
Copyright>
Copyright>        This program is distributed in the hope that it will be useful,
Copyright>        but WITHOUT ANY WARRANTY; without even the implied warranty of
Copyright>        MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
Copyright>        GNU Affero General Public License for more details.
Copyright>
Copyright>        You should have received a copy of the GNU Affero General Public License
Copyright>        along with this program.  If not, see <https://www.gnu.org/licenses/>.
Copyright>
Copyright>
Copyright>        Commercial Alternative: Altair Radioss Software
Copyright>
Copyright>        As an alternative to this open-source version, Altair also offers Altair Radioss
Copyright>        software under a commercial license.  Contact Altair to discuss further if the
Copyright>        commercial version may interest you: https://www.altair.com/radioss/.
Chd|====================================================================
Chd|  SPMD_VFI_SMS                  source/mpi/ams/spmd_vfi_sms.F 
Chd|-- called by -----------
Chd|        SMS_MAV_LT                    source/ams/sms_pcg.F          
Chd|-- calls ---------------
Chd|====================================================================
      SUBROUTINE SPMD_VFI_SMS(V     ,SIZE,VFI,FR_RMS,
     .                       FR_SMS ,LIST_RMS ,LIST_SMS,IFLAG,
     .                       IAD_SEND,IAD_RECV,REQ_R,REQ_S,RBUF,SBUF)
C-----------------------------------------------
C   I m p l i c i t   T y p e s
C-----------------------------------------------
#include      "implicit_f.inc"
C-----------------------------------------------------------------
C   M e s s a g e   P a s s i n g
C-----------------------------------------------
#ifdef MPI
#include "mpif.h"
#endif
C-----------------------------------------------
C   C o m m o n   B l o c k s
C-----------------------------------------------
#include      "com01_c.inc"
#include      "task_c.inc"
C-----------------------------------------------
C   D u m m y   A r g u m e n t s
C-----------------------------------------------
      INTEGER SIZE, IFLAG,
     .        FR_RMS(NSPMD+1), FR_SMS(NSPMD+1),LIST_RMS(*), LIST_SMS(*),
     .        IAD_SEND(NSPMD+1),IAD_RECV(NSPMD+1),
     .        REQ_R(NSPMD),REQ_S(NSPMD)
      my_real
     .        V(SIZE,*), VFI(SIZE,*)
      my_real
     .        RBUF(SIZE*(FR_RMS(NSPMD+1)+FR_SMS(NSPMD+1))),
     .        SBUF(SIZE*(FR_RMS(NSPMD+1)+FR_SMS(NSPMD+1)))
C-----------------------------------------------
C   L o c a l  V a r i a b l e s
C-----------------------------------------------
#ifdef MPI
      INTEGER MSGTYP,I,NOD,LOC_PROC,IERROR,MSGOFF,
     .        SIZ,J,L,NB_NOD,K,M,KK,
     .        STATUS(MPI_STATUS_SIZE)
      DATA MSGOFF/217/
C-----------------------------------------------
C   S o u r c e  L i n e s
C-----------------------------------------------
      LOC_PROC = ISPMD + 1

      IF(IFLAG==1)THEN
        L = 1
        IAD_RECV(1) = 1
        DO I=1,NSPMD

          SIZ = FR_RMS(I+1) - FR_RMS(I)
          IF(I/=LOC_PROC)THEN
            SIZ = SIZ + FR_SMS(I+1) - FR_SMS(I)
          ENDIF
          SIZ = SIZE*SIZ

          IF(SIZ/=0)THEN
            MSGTYP = MSGOFF 
            CALL MPI_IRECV(
     S        RBUF(L),SIZ,REAL,IT_SPMD(I),MSGTYP,
     G        MPI_COMM_WORLD,REQ_R(I),IERROR)
          ENDIF
          L = L + SIZ
          IAD_RECV(I+1) = L

        END DO

        L = 1
        IAD_SEND(1) = 1
        M = 1
        DO I=1,NSPMD
          IF(I/=LOC_PROC)THEN
#include      "vectorize.inc"
            DO J=FR_SMS(I),FR_SMS(I+1)-1
              NOD = LIST_SMS(M)
              IF(NOD/=0)THEN
                DO K=1,SIZE
                  SBUF(L+K-1) =  V(K,NOD)
                END DO
              ELSE
                DO K=1,SIZE
                  SBUF(L+K-1) =  ZERO
                END DO
              END IF
              L = L + SIZE
              M = M + 1
            END DO
          END IF
#include      "vectorize.inc"
          DO J=FR_RMS(I),FR_RMS(I+1)-1
            NOD = LIST_RMS(J)
            IF(NOD/=0)THEN
              DO K=1,SIZE
                SBUF(L+K-1) =  V(K,NOD)
              END DO
            ELSE
              DO K=1,SIZE
                SBUF(L+K-1) =  ZERO
              END DO
            END IF
            L = L + SIZE
          END DO
          IAD_SEND(I+1) = L
        ENDDO
C
C   echange messages
C
        DO I=1,NSPMD
C--------------------------------------------------------------------
          IF(IAD_SEND(I+1)-IAD_SEND(I)>0)THEN
             MSGTYP = MSGOFF 
             SIZ = IAD_SEND(I+1)-IAD_SEND(I)
             L = IAD_SEND(I)
             CALL MPI_ISEND(
     S         SBUF(L),SIZ,REAL,IT_SPMD(I),MSGTYP,
     G         MPI_COMM_WORLD,REQ_S(I),IERROR)
          ENDIF
C--------------------------------------------------------------------
        ENDDO
C
      ELSE ! IFLAG==1
C
        M = 1
        KK= 1
        DO I = 1, NSPMD
          IF(IAD_RECV(I+1)-IAD_RECV(I)>0)THEN
            CALL MPI_WAIT(REQ_R(I),STATUS,IERROR)
            L = IAD_RECV(I)
#include      "vectorize.inc"
            DO J=FR_RMS(I),FR_RMS(I+1)-1
              IF(LIST_RMS(J)/=0)THEN
                DO K=1,SIZE
                  VFI(K,KK) = RBUF(L+K-1)
                END DO
              END IF
              L = L + SIZE
              KK= KK + 1
            END DO
            IF(I/=LOC_PROC)THEN
#include      "vectorize.inc"
              DO J=FR_SMS(I),FR_SMS(I+1)-1
                IF(LIST_SMS(M)/=0)THEN
                  DO K=1,SIZE
                    VFI(K,KK) = RBUF(L+K-1)
                  END DO
                END IF
                L = L + SIZE
                M = M + 1
                KK= KK + 1
              END DO
            END IF
          ENDIF
        END DO
C
        DO I = 1, NSPMD
          IF(IAD_SEND(I+1)-IAD_SEND(I)>0)THEN
            CALL MPI_WAIT(REQ_S(I),STATUS,IERROR)
          ENDIF
        ENDDO
C
      END IF
#endif
      RETURN
      END
